UNITED STATES DEPARTMENT OF THE INTERIOR

GEOLOGICAL SURVEY

BASIC Program to Reduce Audio-magnetotelluric Data and Calculate Apparent Resistivity

by

Carl Long and Herbert A. Pierce

U.S. Geological Survey, Denver, Colorado

Open-File Report 86-200

BASIC Program to Reduce Audio-magnetotelluric Data and Calculate Apparent Resistivity Using Calibration Factors

SUMMARY

This report lists a program for the reduction of orthogonal Audio-magnetotelluric (AMT) soundings that calculates the apparent resistivity for each frequency using calibration factors (K-factors). The program is written in BASIC language for use with the Hewlett-Packard (HP) Model 85 computer and HP 9111A graphics tablet. The program allows input and storage of K-factors, and E and H digitization from field records. It calculates the apparent resistivity, prints these resistivity values, and plots and stores them on either disk or tape.

Advanced Program, Input/Output, Mass Storage, Matrix, and Plotter/Printer read only memory chips (ROMs) are required by the HP85 for this program.

GENERAL OPERATION

The Hewlett Packard (HP) 85 computer must be connected to the HP 9111A graphics tablet. The program AMTRED is loaded from either a HP 5.25" floppy disk or from the HP85's built in magnetic tape reader. The program is generally self-prompting and an example program run is included. If the K-factors have been stored on tape or disk, the user can recall them for the current process. If the K factor values are new or not stored on tape/disk, the user is then prompted to enter the values before the reduction process begins.

Once the K-factors are entered or retrieved from memory the HP 9111A graphics tablet is used to digitize the magnitude of the E and H fields from the field data sheets. Typically, about 10 time-coincident pulses are digitized. The ratios of E to H are calculated and averaged for each of the 16 frequencies. When the 10 ratios have been picked the program displays the ratios for viewing and consequent editing.

Editing is done using arrow keys to position the cursor in front of the value to be deleted. The end line key will delete the value before the apparent resistivity is calculated. To continue after editing, type "000" and press the endline key. The program then calculates the scalar apparent resitivity (ρ_a) described by Cagniard (1953) and by Strangway and others (1973):

$$\rho_a = 1/5f\left|\frac{Ex}{Hy}\right|^2$$

where Ex is in millivolts/km, Hy is in gammas and f is frequency in Hertz. The program then asks if the apparent resistivity for this frequency is OK. If the resistivity value is not reasonable type in "N" and try that frequency again or go on to the next frequency. After going through frequencies in both orthogonal directions (NS/EW) you may print and/or plot, and/or store the station on a mass storage device.

The following is a list of variables that are dimensioned in the program.

Variables	Use
P$[20]	PROJECT NAME
P$[3]	ANSWER FIELD
D4$[20]	LABEL FOR K FACTORS
S5$[6]	STATION IDENTIFICATION
T$	TELLURIC DIRECTION
D4$[6]	K FACTOR LABEL
F$[6]	FILE NAME (TAPE OR DISK)

The following is a sample run and the sample data sheet used for this example:

```
LOAD "AMTRED:0701"
RUN
AMT REDUCTION

PROJ NAME <20 CHAR
?
EXAMPLE RUN

DO YOU HAVE K FACTORS
ON TAPE ? (Y/N)
?
N
Label(<20char),No. of Freq.
ex (SYS-4 10/23/83,16)
?
SYS-6A 1/27/86,16
Input First 4 Freq. 1 at a time
?
4.5
?
7.5
?
14
?
27
ENTER K FACTORS N-S
SMALLEST FREQ TO LARGEST
?
44.7
?
73.96
?
129 71
?
246 16
?
482.34
?
714.66
?
1454 4
?
4603 3
?
460 37
?
1309 6
?
5244.8
?
105 74
?
13.096
?
1 976
?
.182
?
.02
```

COMMENTS

The ":0701" indicates the second floppy disk on a attached mass storage unit. If the HP85 is used without the mass storage unit simply type LOAD "AMTRED" and ENDLINE. The program is then read from the HP85's built in tape device.

This label allows the user to keep track of the system used and date new K-factors were calculated.

This input is required because the older AMT receivers used slightly different frequencies.

Hand input of new K-factors.

```
ENTER K FACTORS E-W
SMALLEST FREQ TO LARGEST
?
54.772
?
79.341
?
133.17
?
255.69
?
495.64
?
768.76
?
1706.3
?
5157.3
?
598.59
?
1578.6
?
5244.8
?
155.25
?
15.478
?
3.868
?
.161
?
.013
STORE K FACTORS ON TAPE (Y/N)
?
Y
ENTER K FAC OUTPUT FILE NAME
?
81225
```

It is a good idea to store your K-factors for future use.

```
K FAC LABEL=MARK 6A

DO YOU WISH TO DIGITIZE(Y/N)
?
Y
IS E-AMP TOP OR BOTTOM CH
ENTER (T/B)
?
T
ENTER STATION NO.
?
TEST 1
```

An "N" allows hand input of frequencies and ρ_a from old data sheets.

```
PROJECT=EXAMPLE RUN
ENTER TELLURIC DIR (NS/EW)
?
NS
ENTER FREQ.
?
27
ENTER E-GAIN,H-GAIN
?
30,18
DIGITIZE LOWER LEFT ZERO LINE

DIGITIZE LOWER RIGHT ZERO LINE

DIGITIZE UPPER
LINE 2ND CHANNEL

DIGITIZE BOT-CH & TOP CH

END DIG 1 PT UPPER RIGHT CORNER

NO OBS= 1

NO OBS= 2

NO OBS= 3

NO OBS= 4

NO OBS= 5

Use Cursor to Delete
To Stop Del or Cont
Type (000 end line)
NO. RATIO E/H
 1 -1.6464
 2 -1.9052
 3 - .7632
 4 -1 9766
 5 -1 1079

 4 -1 9766

000 1 1079
NO. FINAL RATIO E/H
 1 -1 6464
 2 -1 9052
 3 - 7632
 4 -1 1079
FREQ= 27  NS
  AP-RES  N OBS   STD ERR
   25 06     4      10.66

E-GAIN= 30 H-GAIN= 18

IS THIS FREQ. OK? (Y/N)
?
Y
```

The digitizer tablet "beep" prompts the user to enter the limits of the data sheet.

The user begins to digitize the amplitudes of recorded events (see sample of data sheet at end computer example).

The computer counts the number of digitized events. About 10 events are typically digitized for a reasonable average.

The cursor delete option uses the arrow keys (↑↓) to move the cursor to the ratio to be deleted and ENDLINE completes the delete. Type "000" and ENDLINE to display the final ratios and calculate the ρ_a from the average of the remaining ratios.

Here the user should observe the ρ_a and standard error. If the numbers are reasonable and E and H gain correct the user can type "Y" and continue or if incorrect redigitize by typing "N".

```
STATION COMPLETE ? (Y/N)
?
Y
PRINT COMPLETED SOUNDING ? (Y/N)
?
Y
--------------------------------
PROJECT=EXAMPLE RUN
STA. ID_TEST 1 NS  NO FREQ= 1
  FREQ    AP-RES   N OBS    STD ERR
   27.0    25.06      4       10.66
STA. ID_TEST 1 EW  NO FREQ= 0
  FREQ    AP-RES   N OBS    STD ERR
PLOT COMPLETED SOUNDING ? (Y/N)
?
Y
--------------------------------
PROJ= EXAMPLE RUN
```

The print query allows the user to review the two orthognal frequencies for apparent resistivities, number of observations, and standard error on the CRT.

The plot option allows the user to view the plotted apparent resistivities vs. frequencies.




```
STORE SOUNDING ON TAPE ? (Y/N)
?
Y
STA. OUTPUT FILE NAME
?
TEST 1
--------------------------------
PROJ= EXAMPLE RUN
STA#                    TEST 1
FILE NAME=              TEST 1
ANOTHER STATION ? (Y/N)
?
N
END OF JOB
```

After viewing the plot CONT must be typed to continue the program.

The program will write to the available mass storage unit i.e., to tape if no mass storage unit is attached to the HP85.



Figure 1. Sample data record. The user should digitize the time concident transient electric and magnetic high amplitude peaks. The larger peaks do three things: 1) they express the largest received energy, 2) they are above the inherent instrument noise level, and 3) they improve the data confidence by lowering the standard error.



Figure 2. Flow chart of the Basic program AMTRED.

PROGRAM LISTING

```
10 ! AMT REDUCTION BY Carl Long
20 ! U.S.G.S 1-6-84
30 ! CATUION ANY ERRORS SHOULD
40 ! BE BROUGHT TO MY ATTENTION
50 ! GOOD LUCK HAVE FUN
60 OPTION BASE 1
70 DIM K$[25],F$[6],T$[2],P$[20],A$[1],B$[1],D4$[20],S5$[6],J1$[4]
80 K$="BP36,50,3;BP34;BP32;BP30"
90 SHORT F2(25,4),F3(25,4),F4(25,4),F1(25)
100 SHORT E1(25),H1(25),R0(25),K1(2,25),K2(2,25)
110 R1,R3,R4=0
120 CLEAR @ BEEP 80,150
130 DISP "AMT REDUCTION"
140 DISP @ DISP "PROJ NAME <20 CHAR" @ INPUT P$
150 BEEP 50,100 @ DISP
160 DISP "DO YOU HAVE K FACTORS"
170 DISP "ON TAPE ? (Y/N)"
180 INPUT A$[1]
190 IF UPC$(A$[1])="Y" THEN 540
200 IF UPC$(A$[1])#"N" THEN 160
210 DISP "Label(<20char),No. of Freq."
220 DISP "ex.(SYS-4 10/23/83,16)"
230 BEEP 50,160 @ INPUT D4$,Z1
240 BEEP 80,120 @ CLEAR
250 DISP "Input First 4 Freq. 1 at a time"
260 FOR J=1 TO 4
270 INPUT F1(J)
280 NEXT J
290 J1=5 @ O1=10
300 FOR K=1 TO 3
310 FOR I=1 TO 4
320 F1(J1)=F1(I)*O1
330 J1=J1+1
340 NEXT I
350 O1=O1*10
360 NEXT K
370 FOR J=1 TO Z1
380 K1(1,J)=F1(J)
390 K2(1,J)=F1(J)
400 NEXT J
410 DISP "ENTER K FACTORS N-S"
420 DISP "SMALLEST FREQ TO LARGEST"
430 BEEP 50,100
440 FOR I=1 TO Z1
450 INPUT K1(2,I)
460 NEXT I
470 DISP "ENTER K FACTORS E-W"
480 DISP "SMALLEST FREQ TO LARGEST"
```

```
490 BEEP 50,100
500 FOR N=1 TO Z1
510 INPUT K2(2,N)
520 NEXT N
530 GOTO 600
540 CLEAR @ DISP "ENTER K-FAC INPUT FILE NAME"
550 BEEP 50,100 @ INPUT F$
560 ASSIGN# 1 TO F$
570 READ# 1 ; D4$,Z1,K1(,),K2(,)
580 ASSIGN# 1 TO *
590 GOTO 720
600 CLEAR
610 DISP "STORE K FACTORS ON TAPE (Y/N)"
620 BEEP 50,100
630 INPUT A$
640 IF UPC$(A$[1])="N" THEN 740
650 IF UPC$(A$[1])#"Y" THEN 610
660 DISP "ENTER K FAC OUTPUT FILE NAME"
670 BEEP 50,100 @ INPUT F$
680 CREATE F$,4
690 ASSIGN# 1 TO F$
700 PRINT# 1 ; D4$,Z1,K1(,),K2(,)
710 ASSIGN# 1 TO *
720 CLEAR
730 DISP "K FAC LABEL=";D4$
740 PRINT "K FAC LABEL= ";D4$
750 CLEAR @ DISP "DO YOU WISH TO DIGITIZE(Y/N)"
760 INPUT A$
770 IF UPC$(A$[1])="N" THEN 3810
780 IF UPC$(A$[1])#"Y" THEN 750
790 DISP "IS E-AMP TOP OR BOTTOM CH"
800 DISP "ENTER (T/B)"
810 BEEP 50,100
820 INPUT B$
830 CLEAR
840 DISP "ENTER STATION NO."
850 G,G1=1
860 BEEP 50,100
870 INPUT S5$
880 IMAGE "________________________________"
890 PRINT USING 880
900 PRINT "PROJECT=";P$
910 DISP "ENTER TELLURIC DIR (NS/EW)"
920 K3,M2,F0,I1=0
930 BEEP 50,100 @ INPUT T$
940 CLEAR
950 DISP "ENTER FREQ."
960 BEEP 50,100
970 INPUT F0
980 IF UPC$(T$)="EW" THEN 1050
```

```
990 IF UPC$(T$)#"NS" THEN 910
1000 FOR J=1 TO Z1
1010 IF K1(1,J)#F0 THEN 1040
1020 K3=K1(2,J)
1030 GOTO 1110
1040 NEXT J
1050 FOR L=1 TO Z1
1060 IF K2(1,L)#F0 THEN 1090
1070 K3=K2(2,L)
1080 GOTO 1110
1090 NEXT L
1100 IF K3<=0 THEN DISP "TRY  AGAIN" @ GOTO 910
1110 BEEP 50,100
1120 DISP "ENTER E-GAIN,H-GAIN"
1130 INPUT E,H
1140 I1=2^((H-E)/6)^2*K3
1150 CLEAR
1160 DISP "DIGITIZE LOWER LEFT ZERO LINE"
1170 DISP
1180 BEEP 50,100
1190 OUTPUT 706 ;"IN;SG"
1200 OUTPUT 706 ;"OS"
1210 ENTER 706 ; S7
1220 IF BIT(S7,0) THEN 1200
1230 OUTPUT 706 ;"OD"
1240 ENTER 706 ; X,Y
1250 OUTPUT 706 ;"BP80,100,5"
1260 DISP "DIGITIZE LOWER RIGHT ZERO LINE"
1270 DISP
1280 OUTPUT 706 ;"DF;SG"
1290 OUTPUT 706 ;"OS"
1300 ENTER 706 ; S7
1310 IF BIT(S7,0) THEN 1290
1320 OUTPUT 706 ;"OD"
1330 ENTER 706 ; X1,Y1
1340 OUTPUT 706 ;"BP80,100,5;BP80,100,5"
1350 J=0
1360 A1=Y1-Y @ B1=X-X1
1370 C1=-(A1*X)-B1*Y
1380 DISP "DIGITIZE UPPER"
1390 DISP "LINE 2ND CHANNEL"
1400 OUTPUT 706 ;"DF;SG"
1410 OUTPUT 706 ;"OS"
1420 ENTER 706 ; S7
1430 IF BIT(S7,0) THEN 1410
1440 OUTPUT 706 ;"OD"
1450 ENTER 706 ; S1,T1
1460 OUTPUT 706 ;"BP80,100,5;BP80,100,5;BP80,100,5"
1470 D=ABS(A1*S1+B1*T1+C1)/SQR(A1^2+B1^2)/40
1480 CLEAR
```

```
1490 DISP "DIGITIZE BOT-CH & TOP CH"
1500 DISP
1510 DISP "END DIG 1 PT UPPER RIGHT CORNER"
1520 N=1
1530 J=J+1
1540 OUTPUT 706 ;"DF;SG"
1550 OUTPUT 706 ;"OS"
1560 ENTER 706 ; S7
1570 IF BIT(S7,0) THEN 1550
1580 OUTPUT 706 ;"OD"
1590 ENTER 706 ; S,T
1600 OUTPUT 706 ;K$
1610 D1=ABS(A1*S+B1*T+C1)/SQR(A1^2+B1^2)/40
1620 IF N=2 THEN H1(J)=D1-(D-50) @ GOTO 1520
1630 E1(J)=D1
1640 IF E1(J)>52 THEN 1710
1650 N=N+1
1660 J=J-1
1670 CLEAR
1680 DISP "NO OBS=";J+1
1690 IF J>=24 THEN 1720
1700 GOTO 1530
1710 OUTPUT 706 ;"BP120,400,5"
1720 OUTPUT 706 ;"DC"
1730 CLEAR
1740 M2=J-1
1750 IF M2=0 THEN 2540
1760 CLEAR
1770 DISP "Use Cursor to Delete"
1780 DISP "To Stop Del or Cont"
1790 DISP "Type  (000 end line)"
1800 WAIT 2000
1810 DISP "NO. RATIO E/H"
1820 FOR I=1 TO M2
1830 IF UPC$(B$[1])="T" THEN 1860
1840 R0(I)=E1(I)/H1(I)
1850 GOTO 1870
1860 R0(I)=H1(I)/E1(I)
1870 DISP I;R0(I)
1880 NEXT I
1890 D9=CURSROW @ D8=CURSCOL
1900 ALPHA D9-M2,D8
1910 LINPUT "",J1$[1,3]
1920 C=VAL(J1$)
1930 IF C=0 THEN 1960
1940 R0(C)=0
1950 GOTO 1910
1960 L1=1
1970 FOR M=1 TO M2
1980 IF R0(M)=0 THEN 2010
```

```
1990 RO(L1)=RO(M)
2000 L1=L1+1
2010 NEXT M
2020 CLEAR @ M2=L1-1
2030 DISP "NO. FINAL RATIO E/H"
2040 FOR L=1 TO M2
2050 DISP L;RO(L)
2060 NEXT L
2070 R1,R2,R3,R4,R5,N1,E2=0
2080 FOR N=1 TO M2
2090 IF RO(N)=0 THEN DISP "TRY  AGAIN" @ GOTO 910
2100 R1=LOG(RO(N)^2*I1)+R1
2110 R2=RO(N)^2*I1
2120 R3=R2+R3
2130 R4=R2^2+R4
2140 N1=N
2150 NEXT N
2160 R5=EXP(R1/N1)
2170 IF N1<=1 THEN 2190
2180 E2=SQR((R4-R3^2/N1)/(N1-1))/SQR(N1)
2190 IF N1=0 THEN 2540
2200 IF R5>30000 THEN 2450
2210 IF R5<.25 THEN 2500
2220 PRINT "FREQ=";FO;" ";T$
2230 IMAGE "  AP-RES  N OBS   STD ERR"
2240 PRINT USING 2230
2250 PRINT USING 2260 ; R5;N1;E2
2260 IMAGE 5D.DD,6D,8D.DD
2270 IF UPC$(T$)="EW" THEN 2320
2280 F2(G,1)=FO @ F2(G,2)=R5
2290 F2(G,3)=N1 @ F2(G,4)=E2
2300 G=G+1
2310 GOTO 2350
2320 F3(G1,1)=FO @ F3(G1,2)=R5
2330 F3(G1,3)=N1 @ F3(G1,4)=E2
2340 G1=G1+1
2350 DISP
2360 DISP "E-GAIN=";E;"H-GAIN=";H
2370 DISP
2380 DISP "IS THIS FREQ. OK? (Y/N)"
2390 INPUT A$
2400 IF UPC$(A$[1])="Y" THEN 2540
2410 IF UPC$(A$[1])#"N" THEN 2380
2420 IF UPC$(T$)="NS" THEN G=G-1
2430 IF UPC$(T$)="EW" THEN G1=G1-1
2440 GOTO 2540
2450 PRINT "AP-RES TOO LARGE";R5
2460 BEEP 50,100 @ BEEP 60,100 @ BEEP 70,100
2470 DISP " RES TOO LARGE"
2480 DISP " TRY AGAIN OR CONTINUE"
```

```
2490 GOTO 2540
2500 PRINT "AP-RES TOO SMALL";R5
2510 BEEP 50,100 @ BEEP 60,100 @ BEEP 70,100
2520 DISP " RES TOO SMALL"
2530 DISP " TRY AGAIN OR CONTINUE"
2540 BEEP 50,100
2550 DISP
2560 DISP
2570 DISP
2580 DISP "STATION COMPLETE ? (Y/N)"
2590 INPUT A$
2600 CLEAR
2610 IF UPC$(A$[1])="N" THEN 910
2620 IF UPC$(A$[1])#"Y" THEN 2570
2630 ! SORT FREQ N-S
2640 FOR K=1 TO G-2
2650 FOR M=K+1 TO G-1
2660 IF F2(K,1)<=F2(M,1) THEN 2720
2670 FOR I=1 TO 4
2680 F4(K,I)=F2(K,I)
2690 F2(K,I)=F2(M,I)
2700 F2(M,I)=F4(K,I)
2710 NEXT I
2720 NEXT M
2730 NEXT K
2740 ! SORT FREQ E-W
2750 FOR K=1 TO G1-2
2760 FOR M=K+1 TO G1-1
2770 IF F3(K,1)<=F3(M,1) THEN 2830
2780 FOR I=1 TO 4
2790 F4(K,I)=F3(K,I)
2800 F3(K,I)=F3(M,I)
2810 F3(M,I)=F4(K,I)
2820 NEXT I
2830 NEXT M
2840 NEXT K
2850 CLEAR
2860 BEEP 114,220
2870 DISP "PRINT COMPLETED SOUNDING ? (Y/N)"
2880 INPUT A$
2890 IF UPC$(A$[1])="N" THEN 3100
2900 PRINT
2910 PRINT USING 880
2920 PRINT "PROJECT=";P$
2930 PRINT
2940 PRINT "STA. ID_";S5$;" NS  NO FREQ=";G-1
2950 PRINT
2960 PRINT USING 2970
2970 IMAGE "  FREQ   AP-RES  N OBS   STD ERR"
2980 FOR I=1 TO G-1
```

```
2990 PRINT USING 3000 ; F2(I,1);F2(I,2);F2(I,3);F2(I,4)
3000 IMAGE 5D.D,5D.DD,6D,8D.DD
3010 NEXT I
3020 PRINT
3030 PRINT "STA. ID_";S5$;" EW  NO FREQ=";G1-1
3040 PRINT
3050 PRINT USING 2970
3060 FOR I=1 TO G1-1
3070 PRINT USING 3000 ; F3(I,1);F3(I,2);F3(I,3);F3(I,4)
3080 NEXT I
3090 PRINT
3100 BEEP 100,300
3110 DISP "PLOT COMPLETED SOUNDING ? (Y/N)"
3120 INPUT A$
3130 IF UPC$(A$[1])="N" THEN 3620
3140 PRINT
3150 PRINT
3160 PRINT USING 880
3170 PRINT "PROJ= ";P$
3180 PLOTTER IS 1
3190 GCLEAR
3200 FRAME
3210 SCALE -1,6.5,-.5,5.5
3220 XAXIS 1,-1,1,6
3230 YAXIS 1,1,1,5
3240 FOR M=1 TO 6
3250 LDIR 0
3260 MOVE -M+6.75,.4
3270 LABEL "10"
3280 MOVE -M+7.2,.61
3290 LABEL VAL$(M-1)
3300 NEXT M
3310 FOR R=1 TO 5
3320 MOVE .2,R-1+.9
3330 LABEL "10"
3340 MOVE .7,R-1+1.1
3350 LABEL VAL$(R-1)
3360 NEXT R
3370 FOR M=1 TO G-1
3380 MOVE -LGT(F2(M,1))+6,LGT(F2(M,2))+1
3390 LABEL "o"
3400 NEXT M
3410 FOR U=1 TO G1-1
3420 MOVE -LGT(F3(U,1))+6,LGT(F3(U,2))+1
3430 LABEL "x"
3440 NEXT U
3450 MOVE 2,-.25
3460 LABEL " FREQUENCY"
3470 LDIR 90
3480 MOVE -.5,2
```

```
3490 LABEL "APPARENT"
3500 MOVE -.2,1.5
3510 LABEL "RESISTIVITY"
3520 LDIR 0
3530 MOVE -.5,.3
3540 LABEL "O=NS"
3550 MOVE -.5,0
3560 LABEL "X=EW"
3570 ALPHA
3580 MOVE -.5,-.3
3590 LABEL "STA# ";S5$
3600 PRINT
3610 COPY
3620 BEEP 120,250
3630 DISP "STORE SOUNDING ON TAPE ? (Y/N)"
3640 INPUT A$
3650 IF UPC$(A$[1])="N" THEN 3770
3660 DISP "STA. OUTPUT FILE NAME"
3670 N6=G-1 @ N7=G1-1
3680 INPUT F$
3690 CREATE F$,8
3700 ASSIGN# 1 TO F$
3710 PRINT# 1 ; F$,S5$,N6,N7,F2(,),F3(,)
3720 ASSIGN# 1 TO *
3730 PRINT
3740 PRINT USING 880
3750 PRINT "PROJ= ";P$
3760 PRINT "STA# ",S5$,"FILE NAME=",F$
3770 BEEP 160,200
3780 DISP "ANOTHER STATION ? (Y/N)"
3790 INPUT A$
3800 IF UPC$(A$[1])="Y" THEN 830
3810 CLEAR
3820 FOR I=1 TO 10
3830 BEEP I*RND+1,200
3840 NEXT I
3850 DISP "END OF JOB"
3860 DISP
3870 DISP "TAKE A BREAK"
3880 DISP
3890 DISP "REMOVE TAPE"
3900 DISP
3910 DISP "TURN OFF HP-85 &DIGITIZER"
3920 END
```

REFERENCES

Cagniard, L., 1953, Basic theory of magnetotelluric method of geophysical prospecting: Geophysics, v. 18, p.605-635.

Strangway, D. W., Swift, C. M., Jr., and Holmer, R. C., 1973: The application of audio-frequency magnetotelluric (AMT) to mineral exploration: Geophysics, v. 38, p. 1159-1175.